query: embed.html##-----------------------------------------------------
  #Viewpoint
  #-----------------------------------------------------
  PREFIX schema: <http://schema.org/>
  #defaultView:Map{"hide":["?coordinate", "?line", "?rgb", "?lon1", "?lat1", "?lon2", "?lat2"]}
  SELECT distinct ?item ?itemLabel ?coordinate ?coordinate2 ?image ?lon1 ?lat1 ?lon2 ?lat2 ?line ?distNum ?sitelink
  
  #####################################################################
  WITH {
    # Liste der POIs, die man vom Aussichtspunkt sehen kann
  SELECT ?item
         
         (SAMPLE(?coordinate) AS ?coordinate)
         #(SAMPLE(?coordinate2) AS ?coordinate2)
         (SAMPLE(?lon1) AS ?lon1)
         (SAMPLE(?lat1) AS ?lat1)
         (SAMPLE(?lon2) AS ?lon2)
         (SAMPLE(?lat2) AS ?lat2)
         (CONCAT(STR(ROUND(SAMPLE(?distNum)*100)/100),' km') AS ?distNum)  # Entfernung runden
         (SAMPLE(?sitelink) AS ?sitelink)
         (SAMPLE(?itemLabel) AS ?itemLabel) 
         (SAMPLE(?image) AS ?image)
         
    WHERE {
    VALUES (?poi) {
      (wd:Q277572)     # Fernsehturm Dresden
      (wd:Q322853)     # Meißner Dom
      (wd:Q866559)    # Bismarckturm Radebeul
      (wd:Q894173)     # Borsberg
      (wd:Q881976)     # Großer Winterberg 
      (wd:Q700562)     # Festung Königstein
      (wd:Q227259)     # Zschirnsteine 
      (wd:Q453233)     # Pfaffenstein  
      (wd:Q6811)    # Windberg
      (wd:Q1272672)    # Hoher Schneeberg 
      (wd:Q1136672)    # Cottaer Spitzberg 
      (wd:Q880766)     # Geisingberg 
      (wd:Q31310959)   # Lugstein  
      (wd:Q1720762)    # Kahleberg  
      (wd:Q1867648)    # Lockwitztalbrücke
      (wd:Q1600678)    # Wilisch 
      (wd:Q797717)     # Babisnauer Pappel 
      (wd:Q1248323)    # Gohlig 
      (wd:Q1679651)    # Panometer Dresden  
      (wd:Q1231293)    # Marienschacht  
      (wd:Q1087301)    # Christuskirche Strehlen 
      (wd:Q2358380)    # Studentenwohnheime Wundtstraße
      (wd:Q22949130)   # Tunnel Altfranken
      (wd:Q2409158)    # Tharandter Wald
      (wd:Q827406)     # Forstbotanischer Garten Tharandt
      (wd:Q1801966)    # Landberg
  
  
   
  
    }	    
    BIND(?poi as ?item) .
    
      # Koordinate von POIs auseinander nehmen
    ?item wdt:P625 ?coordinate  .
    OPTIONAL{?item p:P625 / psv:P625 / wikibase:geoLatitude ?lat2 . }
    OPTIONAL{?item p:P625 / psv:P625 / wikibase:geoLongitude ?lon2 . }
    OPTIONAL{?item wdt:P18 ?image }.
  
    ?sitelink schema:about ?item.
    FILTER(REGEX(STR(?sitelink), "de.wikipedia.org/wiki/"))    
      
      
    # Koordinate vom Ausgangspunkt (manuelle Eingabe)
    #BIND("51.03402" as ?lat1) . 
    #BIND("13.80267" as ?lon1) .
    #BIND(CONCAT("Point(",str(?lon1)," ",str(?lat1),")") AS ?string1) .
    #BIND(STRDT(?string1, geo:wktLiteral) AS ?coordinate2).
      
    # Koordinate vom Ausgangspunkt (per Wikidata-Objekt) Mimosa AG ()  
    BIND(wd:Q15296697 as ?viewpoint) .
    ?viewpoint wdt:P625 ?coordinate2  .
    OPTIONAL{?viewpoint p:P625 / psv:P625 / wikibase:geoLatitude ?lat1 . }
    OPTIONAL{?viewpoint p:P625 / psv:P625 / wikibase:geoLongitude ?lon1 . }
    OPTIONAL { ?viewpoint wdt:P18 ?image }.
      
    BIND(CONCAT('LINESTRING (', STR(?lon1), ' ', STR(?lat1), ',', STR(?lon2), ' ', STR(?lat2), ')') AS ?str) .
    BIND(STRDT(?str, geo:wktLiteral) AS ?line) 
      
    #Entfernung berechnen (geht aktuell nicht bei manueller Koordinateneingabe, unklar warum)
    BIND(geof:distance(?coordinate, ?coordinate2) AS ?distNum).
    
  }    
  GROUP BY ?item
  
  } AS %points
  
  #########################################
  WITH {
    # Linien generieren 
    SELECT ?line ?item ("9B0058" as ?rgb) 
            WHERE {
      INCLUDE %points 
      BIND(CONCAT('LINESTRING (', STR(?lon1), ' ', STR(?lat1), ',', STR(?lon2), ' ', STR(?lat2), ')') AS ?str) .
      BIND(STRDT(?str, geo:wktLiteral) AS ?line) 
      
    }
  } AS %lines 
  #########################################
  
  WHERE {
    
    # jetzt Punkte und Linien für Ausgabe auf Karte ausgeben
   { INCLUDE %points }
   union
   { INCLUDE %lines }
    
    SERVICE wikibase:label { bd:serviceParam wikibase:language "[AUTO_LANGUAGE],de,en" }
  }